

MI PRIMERA BIBLIOTECA INFORMATICA

BASIC FACIL



EDICIONES GENERALES ANAYA



MI PRIMERA BIBLIOTECA INFORMATICA


BASIC FACIL

Gaby Waters

Ilustrado por Graham Round

Programas: Paul Shreeve


Editora de la serie: Helen Davies



EDICIONES GENERALES ANAYA

Contenido

| | | | |
|----|----------------------------|----|-------------------------------|
| 3 | En este libro | 22 | Comparaciones |
| 4 | ¿Qué es BASIC? | 24 | Programas con IF y THEN |
| 6 | Primeros comandos en BASIC | 26 | Bucles |
| 8 | Cómo escribir programas | 28 | Programas con bucles |
| 10 | Introducción de programas | 30 | Números aleatorios |
| 12 | Programas con PRINT | 32 | Cómo depurar programas |
| 14 | Introducción de variables | 33 | Tabla de comandos |
| 16 | Más sobre variables | 34 | Programas y rompecabezas |
| 18 | Uso de INPUT | 44 | Respuestas a los rompecabezas |
| 20 | Bromas con PRINT | 48 | Indice |




En este libro

Este libro es una guía para los que empiezan a escribir programas en BASIC. BASIC es el lenguaje que usan la mayor parte de los ordenadores personales. Es bastante fácil de aprender y, una vez que has dominado unos pocos comandos puedes hacer un montón de cosas divertidas con el ordenador.



En las próximas páginas puedes aprender a programar ordenadores. Posteriormente, se explican cada uno de los principales comandos BASIC, con programas sencillos que muestran su función. Hay además algunos programas divertidos para realizar y rompecabezas de programación para resolver.



Uno o dos comandos BASIC varían de una marca a otra de ordenadores. Están claramente marcados. En la página 33 hay una tabla que muestra la versión que usas en tu ordenador.



Al final del libro aparecen las respuestas a los rompecabezas con explicaciones sencillas. Hay también avisos y sugerencias para ayudarte si tus programas no funcionan.

¿Qué es BASIC?

Para que un ordenador realice algo necesitas darle instrucciones en un lenguaje que entienda. BASIC, el lenguaje que usan la mayor parte de los ordenadores personales, es bastante parecido al inglés, con algunas palabras y símbolos especiales.



Si das instrucciones al ordenador en lenguaje ordinario, no te entenderá.



Dar instrucciones al ordenador se llama programar; una lista de instrucciones se denomina programa. En la pantalla superior puedes ver cómo es un programa BASIC.



El BASIC puede usarse para que el ordenador haga toda clase de cosas, por ejemplo, dibujar, resolver problemas, e incluso ruidos divertidos.

Cómo trabajan los ordenadores



Cuando introduces un programa en el ordenador, éste tiene que traducirlo a código máquina. El

código máquina está formado por señales eléctricas. Los ordenadores usan este código para hacer todo

su trabajo. Puedes escribir programas en código máquina, pero es muy difícil.

Otros lenguajes de programación



Hay montones de lenguajes de programación además del BASIC. El dibujo muestra el lenguaje LOGO, que usa comandos sencillos, como LEFT y RIGHT, para que un ordenador dibuje

contornos en la pantalla. LOGO se usa también para controlar un pequeño robot llamado tortuga. Este tiene un lápiz que dibuja contornos al moverse.



Las instrucciones del dibujo están escritas en un lenguaje de programación llamado FORTRAN, muy útil para resolver problemas matemáticos y científicos.

Primeros comandos en BASIC

En la mayoría de los ordenadores se teclean los comandos BASIC letra a letra. En algunos, sin embargo, basta con pulsar una tecla.

En esta página y la siguiente puedes aprender lo relativo al comando PRINT, que hace que el ordenador escriba algo en la pantalla.



Cuando enciendes el ordenador, un símbolo intermitente llamado cursor aparece en la pantalla. Indica que el ordenador está preparado para recibir instrucciones.



Para hacer que un ordenador escriba una palabra en la pantalla, debes teclear PRINT, seguido de la palabra entre comillas. Teclea cuidadosamente para no confundir al ordenador.



Para que el ordenador ejecute la orden, pulsa la tecla marcada RETURN (o ENTER en algunos ordenadores)



Cuando pulsas RETURN (o ENTER), el ordenador pone la palabra en la pantalla*. Trata de hacer que tu ordenador escriba palabras usando PRINT.


Escritura de mensajes

Más abajo puedes encontrar cómo borrar la pantalla después de cada mensaje.




Aquí hay algunas ideas para escribir mensajes en la pantalla. Usa un comando PRINT para cada mensaje y teclea las palabras entre comillas

Para borrar la pantalla



El comando para borrar la pantalla varía de una marca a otra de ordenadores. En la página 33 puedes encontrar el usado por tu ordenador*.



Estos son algunos de los comandos empleados para borrar la pantalla.

Cuando pulsas RETURN después de este comando el ordenador borra la pantalla excepto el cursor.

Uso de PRINT con números

PRINT 1234

PRINT 999

PRINT 12

PRINT también hace que el ordenador escriba números. No necesitas poner los números entre comillas. Prueba con comandos como los indicados arriba para hacer que el ordenador escriba números.

* Siempre que veas CLS en este libro, utiliza la orden que borra la pantalla en tu ordenador.

Cómo escribir programas

Imagínate tratando de hacer que un ordenador fantasma compre un helado. Una instrucción como la del dibujo sería demasiado difícil para el cerebro del ordenador fantasma. Necesitas darle una lista de comandos llamada programa.

Antes de escribir el programa, piensa lo que el ordenador necesita hacer y escribe un plan.



PLAN

1. Ir al puesto de helados
2. Pedir un helado grande de fresa
3. Pagar al heladero
4. Coger el helado (no comerlo)
5. Dar la vuelta y regresar
6. Entregar el helado a Viscosillo



Tienes que pensar exactamente lo que el ordenador fantasma necesita hacer en cada etapa y hacer una lista de instrucciones paso a paso.

Planificar un programa

PLAN

1. Limpiar la pantalla
2. Preguntar por la clave
3. Esperar respuesta
4. Si la respuesta es correcta, escribir el mensaje BIEN HECHO
5. Si la respuesta es equivocada, escribir un mensaje adecuado.



Cuando escribas un programa, primero planifica exactamente lo que quieres que el ordenador haga. A la izquierda hay un plan para un programa que pregunta por una palabra clave.

← Este comando es diferente en algunos ordenadores. Mira la página 33.

10 CLS PRINT "CUAL ES LA CLAVE?"

30 INPUT P\$

40 IF P\$="SALCHICHAS" THEN PRINT "BIEN HECHO"

50 IF P\$<>"SALCHICHAS" THEN PRINT "FALLASTE , MONSTRUO ESTUPIDO"

Más adelante podrás descubrir lo que significan los comandos BASIC de este programa.

Los números de las líneas usualmente van de diez en diez, de esta forma puedes intercalar comandos posteriormente.

Para pasar el plan a un programa, tienes que traducir cada instrucción a BASIC. Pon cada instrucción en una línea, comenzando con un número de línea. Los números indican al ordenador el orden de las instrucciones.



Introducción de programas

Cuando introduces un programa en el ordenador, éste almacena todos los comandos en su memoria, pero no los ejecuta hasta que no le indiques que lo haga.

A la derecha hay un programa para hacer que un ordenador escriba una poesía tonta. Tecléalo en tu ordenador usando las instrucciones de abajo como ayuda.

En la mayoría de los ordenadores el número 0 está atravesado por una línea para evitar confundirlo con la letra O.

Programa poético

Recuerda que este comando es distinto en algunos ordenadores.

```
10 CLS
20 PRINT "LAS ROSAS SON ROJAS"
30 PRINT "LAS VIOLETAS SON AZULES"
40 PRINT "CON UNA CARA COMO ESA"
50 PRINT "ESTARIAS MEJOR EN UNA JAULA"
```

El programa usa el comando PRINT para hacer que el ordenador escriba las palabras de la poesía.

Cómo teclear un programa




Debes teclear el número al comienzo de cada línea.

Teclea el programa línea a línea. Al final de cada línea comprueba que no haya errores y luego pulsa RETURN. Esto hace al ordenador almacenar la línea en su memoria.



La tecla DELETE puede tener un nombre distinto en tu ordenador. Compruébalo en la página 33 o en tu manual.


Si cometes un error al teclear, puedes borrarlo pulsando la tecla DELETE (a menudo marcada con la abreviatura DEL).



Cuando ejecutes el programa la pantalla deberá quedar como ésta.

Para hacer que el ordenador ejecute el programa, teclea el comando RUN y pulsa RETURN. Esto se denomina correr el programa.

Errores en programas



El ordenador puede escribir un mensaje de error como éste.

Si un programa no se ejecuta correctamente, probablemente has introducido algún error. Un error al teclear puede parar la ejecución de un programa. En inglés, los errores en los programas se denominan «bugs». Para corregirlos, mira la página 32.

Cómo borrar programas



Algunos ordenadores borran la pantalla cuando tecleas NEW.

Para borrar un programa, teclea NEW y pulsa RETURN. Esto hace desaparecer el programa de la memoria del ordenador.

Más sobre programas



Puedes teclear las líneas del programa en el orden que quieras. Prueba a teclear este programa y a ejecutarlo. El ordenador clasificará las líneas y las ejecutará en orden.

Programa de fiesta de cumpleaños

```
30 PRINT "POR FAVOR VEN A MI"  
70 PRINT "BONITO REGALO"  
60 PRINT "Y TRAEME UN"  
20 PRINT "QUERIDO UGLYMUG"  
50 PRINT "EL SABADO"  
10 CLS  
80 PRINT "CON CARÍÑO VISCOSILLO"  
40 PRINT "FIESTA DE CUMPLEAÑOS"
```

También puedes hacer que el ordenador escriba una lista de las líneas del programa en el orden correcto. Para hacerlo teclea LIST y pulsa RETURN a continuación.

Para cambiar una línea basta con teclear la nueva versión con el mismo número de línea.

Par borrar una línea del programa teclea el número de la línea y pulsa RETURN.

```
20  
70 PRINT "UN PASTEL DE CHOCOLATE"  
75 PRINT "<ME GUSTA EL CHOCOLATE>"
```

Puedes añadir líneas en cualquier lugar del programa. El número de la línea indica al ordenador dónde colocarla.

Es muy fácil cambiar programas añadiendo, alterando o borrando líneas. Prueba a hacer los cambios mostrados arriba en el programa de fiesta de cumpleaños.

Programas con PRINT

Puedes usar PRINT para dibujar formas y figuras a base de símbolos. El programa de la derecha traza el contorno de un cohete usando símbolos, letras y espacios. Ensáyalo en tu ordenador.

Programa cohete

```
10 CLS
20 PRINT "□□□!"
30 PRINT "□□!□!"
40 PRINT "□!□!□!"
50 PRINT "□!□!□!"
60 PRINT "□!□!□!"
70 PRINT "□!□!□!"
80 PRINT "XXX□XXX"
```

Este símbolo quiere decir que teclees un espacio. Por cada □ debes pulsar una vez la barra de espacios.

Cada comando PRINT dibuja una línea de símbolos y espacios. Los símbolos y espacios deben estar entre comillas.

Recuerda usar el comando para borrar la pantalla utilizado por tu ordenador.

Programa para dibujar una cara

1

```
***
*****
*****
1 1
1 o o 1
( , )
1 1
1 o 1
```

2

10 CLS

Pon un comando para borrar la pantalla al principio del programa.

3

Puedes comprobar tu programa en las respuestas de la página 44.

Prueba a escribir un programa para hacer que tu ordenador dibuje una cara como la mostrada arriba.

En el programa necesitarás un comando PRINT para cada línea del dibujo. Puede servirte de ayuda el dibujar primero la cara en papel cuadriculado. Así pueden averiguar cuántos símbolos y espacios necesitas en cada línea.

Cómo hacer operaciones



Haz unas cuantas operaciones para que las resuelva el ordenador.

El símbolo * significa para el ordenador multiplicar, y / significa dividir. Usa los signos normales para la suma y la resta.

El comando PRINT, seguido de una suma, indica al ordenador que efectúe la suma y coloque el resultado en la pantalla. Prueba con los comandos PRINT mostrados a la izquierda. Los comandos no tienen número de línea, de esta forma el ordenador realiza cada comando directamente.

Programa para calcular horas

```
10 CLS  
20 PRINT "HAY"  
30 PRINT 365*24  
40 PRINT "HORAS EN"  
50 PRINT "UN AÑO"
```

Este programa hace que el ordenador calcule el número de horas de un año. La línea 30 hace el cálculo. Multiplica el número de días de un año por el número de horas de un día. Prueba a ejecutar el programa en tu ordenador.

24 es el número de horas de un día.

365 es el número de días de un año.

Esto es lo que aparecerá en tu pantalla cuando ejecutes el programa.



Introducción de variables

Cuando escribes programas, a menudo necesitas dar información al ordenador para que trabaje. El ordenador almacenará esta información en celdas, llamadas variables, en su memoria. Una forma de poner información en el ordenador es mediante el comando LET.



Las celdas en la memoria del ordenador son parecidas a las cajas vacías del dibujo. Cuando das información al ordenador, necesitas darle también una etiqueta. El ordenador pone la información en una celda y la etiqueta. Una celda etiquetada se denomina variable.

Uso de LET

Este es el nombre de la variable.


LET N=999

Este es el número almacenado en la variable.

El comando LET de arriba indica al ordenador que almacene el número 999 en una variable llamada N.



Rompecabezas LET



Prueba a escribir en un papel tres comandos LET para almacenar los números en las variables mostradas arriba. Si tienes dudas, mira las respuestas en la página 44.



En BASIC, las palabras, letras y símbolos se denominan hileras. Cuando almacenas una hilera en una variable debes teclear la hilera entre comillas. Una variable que contiene una hilera se denomina variable hilera. Su nombre siempre tiene el signo del dólar al final.



Uso de PRINT con variables

```
10 CLS
20 LET N=99
30 LET B$="PLATANOS"
40 PRINT N
50 PRINT B$
```



Los comandos PRINT le dicen al ordenador que visualice el contenido de las variables N y B\$.

Para que el ordenador escriba la información almacenada en una variable, usa PRINT seguido del nombre de la variable. Prueba a ejecutar el programa de arriba para comprobar cómo funciona.

Rompecabezas PRINT

```
10 CLS
20 LET A$="EL CEREBRO"
30 LET B$="ES"
40 LET C$="ABURRIDO"
```




Prueba a añadir tres comandos PRINT a este programa para hacer que el ordenador escriba la información almacenada en A\$, B\$ y C\$. La respuesta está en la página 44.

Más sobre variables

En esta página y en la próxima puedes aprender más cosas sobre nombres de variables y cómo usarlas en programas.

Nombres de variables



Es una buena idea dar a una variable un nombre que te ayude a recordar lo que contiene.

Las letras del alfabeto se usan normalmente como nombres de variables. En la mayor parte de los ordenadores, sin embargo, puedes usar una letra seguida de un número e incluso una palabra corta. El manual de tu ordenador te indicará los nombres que permite usar.

Nombres ilegales



Debes tener cuidado de no usar nombres que contengan comandos BASIC como los mostrados. Ello confundirá al ordenador y conducirá a fallos en el programa.

Palabras y variables juntas

1

```
10 CLS  
20 LET N=7  
30 PRINT N;" BABOSILLO"  
40 PRINT "COMIO";N;"BOCADILLOS PASTOSOS"
```

Esto es un punto y coma.

Estos espacios sirven para separar las palabras y el contenido de la variable.

A menudo necesitas visualizar palabras junto al contenido de una variable. Para hacer esto, usa

PRINT seguido del nombre de la variable y las palabras separadas por punto y coma. En BASIC, un

2

```
7 "BABOSILLO"  
COMIO 7 BOCADILLOS PASTOSOS
```

Esto aparecerá en tu pantalla cuando ejecutes tu programa.

punto y coma le indica al ordenador que escriba lo que sigue en la misma línea.

Operaciones con variables

1

```
10 CLS
20 LET A=2
30 LET B=5
40 PRINT A+B
50 PRINT A*B
```

Estas son las respuestas a las operaciones.



2

```
35 PRINT A;"+";B;"=";
45 PRINT A;"x";B;"=";
```

2+5=7
2x5=10

Puedes también usar el punto y coma al final de una línea para indicar al ordenador que no comience una nueva línea.



Puedes hacer que el ordenador realice operaciones con números almacenados en variables. Prueba con el programa de arriba. Este hace que el ordenador sume y multiplique los números almacenados en las variables A y B.

Las respuestas escritas en la pantalla no resultan demasiado explicativas. Si añades estas dos líneas PRINT al programa, puedes hacer que el ordenador escriba la operación y la respuesta.

Programa del pastel


```
10 CLS
20 LET C=12
30 LET M=4
40 LET A=C/M
50 PRINT "HAY ";A;" PASTELES"
60 PRINT "PARA CADA MONSTRUO"
```

Ejecuta el programa para descubrir cuántos pasteles tocan a cada monstruo.


Cuatro monstruos tienen 12 pasteles y quieren repartirlos equitativamente. El programa de arriba resuelve el problema de los monstruos. La variable C contiene el número de pasteles. M es el número de monstruos y A la respuesta.

Uso del INPUT

Otra forma de dar información al ordenador es con el comando INPUT. INPUT te permite poner información en la memoria del ordenador mientras el programa se está ejecutando. Esto significa que puedes dar distinta información al ordenador cada vez que ejecutes un programa. A la derecha puedes descubrir cómo funciona el comando INPUT.



Coloca siempre el nombre de una variable después de INPUT. Cuando el ordenador encuentra INPUT pone una etiqueta en una celda de la memoria y espera que teclees algo.



Escribiendo una interrogación, el ordenador te indica que espera que teclees algo. Cuando tecleas la información, el ordenador la almacena.

Un programa con INPUT

El comando PRINT escribe la pregunta.

```
10 CLS
20 PRINT "COMO TE LLAMAS?"
30 INPUT N$
40 PRINT "QUE EDAD TIENES?"
50 INPUT A
60 CLS
70 PRINT N$; "TIENE ";A
80 PRINT "AÑOS"
```

Esta línea le indica al ordenador que se esperan palabras.

Esta línea le indica al ordenador que se espera un número.

Aquí hay un programa de prueba. Usa INPUT para permitirte teclear tu nombre y edad. Después el ordenador escribe esta información en la pantalla.

Puedes ejecutar el programa varias veces dando diferentes nombres y edades. Prueba con nombres y edades sin sentido.

Calculadora con ordenador



Necesitas almacenar los números en dos variables llamadas A y B.

Esta línea le dice al ordenador que sume los dos números y almacene la respuesta en C.

```
10 CLS
20 PRINT "TECLEA UN NUMERO"
30
40 PRINT "TECLEA OTRO NUMERO"
50
60 LET C=A+B
70 PRINT A;"+";B;"=";C
```

Prueba a rellenar las líneas vacías del programa. Este programa hace trabajar a tu ordenador como una calculadora. Teclea dos números y el ordenador

los suma y coloca la respuesta en la pantalla. Necesitas rellenar las líneas 30 y 50 con comandos INPUT. Puedes ver la respuesta en la página 44.

Rompecabezas de programa mensaje

```
10 CLS
20 PRINT "CUAL ES EL MENSAJE?"
30 INPUT M$
40 PRINT "PARA QUIEN ES?"
50 INPUT A$
60 PRINT "DE QUIEN ES?"
70 INPUT B$
80 CLS
90 PRINT
100 PRINT M$
110 PRINT
```

Estos son los comandos que necesitas completar. Puedes comprobar tu respuesta en la página 44.

Cuando ejecutas el programa, el ordenador pregunta por un mensaje. Luego pregunta para quién es el mensaje y quién lo envía.

El ordenador almacena esta información en las variables AS, BS y MS. Luego borra la pantalla y visualiza el mensaje.

¿Puedes completar los comandos PRINT de las líneas 90 y 110 para que el ordenador escriba los nombres almacenados en AS y BS con el mensaje?



Bromas con PRINT

En estas dos páginas puedes encontrar cómo escribir cosas en distintos lugares de la pantalla usando PRINT.

PRINT TAB



Prueba a cambiar los números que siguen a TAB y comprueba lo que ocurre.

```
10 CLS
20 PRINT TAB(5);"YO SOY GRANDE"
30 PRINT TAB(8);"PERO VISCOSILLO"
40 PRINT TAB(12);"ES ENANO"
```



El comando PRINT TAB le dice al ordenador que deje algunos espacios en la pantalla antes de

una palabra o número. Le indicas al ordenador cuántos espacios debe dejar poniendo un

número entre paréntesis después de TAB. Trata de ejecutar el programa para comprobarlo.

Programa con PRINT TAB



```
10 CLS
20 PRINT "CUANTOS ESPACIOS"
30 PRINT "QUIERES DEJAR?"
40 INPUT N
50 PRINT TAB(N);"HOLA"
```

En este programa le dices al ordenador cuántos espacios debe dejar antes de la palabra HOLA. El ordenador almacena el número

en la variable N y luego usa N con el comando PRINT TAB. Ejecuta el programa varias veces dando cada vez un número distinto.

Cómo dejar líneas en blanco

```
10 CLS
20 PRINT "HOLA"
30 PRINT
40 PRINT
50 PRINT
60 PRINT "ADIOS"
```

Para que el ordenador deje líneas en blanco en la pantalla usa el comando PRINT sin nada a continuación. Este programa hace que el ordenador deje tres líneas en blanco entre los dos mensajes.

Rompecabezas de programa de cumpleaños

Esto son dos puntos.

Puedes poner varios comandos en una línea si los separas con dos puntos.

```
10 CLS
20 PRINT:PRINT:PRINT
30 PRINT:PRINT:PRINT
40 PRINT TAB( );"FELIZ"
50 PRINT TAB( );"CUMPLEAÑOS"
60 PRINT TAB( );"*****"
```

Este programa escribe un mensaje de cumpleaños. Antes de ejecutarlo debes poner los números en los paréntesis que siguen a los comandos PRINT TAB. Experimenta con números distintos para tratar de poner el mensaje en el centro de la pantalla, según se muestra a la derecha.



Cómo hacer columnas

Prueba a ejecutar este programa. Esto es lo que debe aparecer en tu pantalla.

```
10 CLS
20 PRINT:PRINT:PRINT
30 PRINT "DIA","MI COMIDA"
40 PRINT
50 PRINT "LUNES","CHOCOLATE"
60 PRINT "MARTES","HAMBURGUESAS"
70 PRINT "MIERCOLES","HELADO"
80 PRINT "JUEVES","SALCHICHAS"
90 PRINT "VIERNES","POLOS"
100 PRINT "SABADO","PATATAS"
110 PRINT "DOMINGO","PALOMITAS"
```



Para hacer que el ordenador escriba palabras o números en columnas, usa comas.

Una coma hace que el ordenador ponga la siguiente palabra en una nueva columna.

Comparaciones

Puedes hacer que un ordenador compare informaciones y realice cosas distintas de acuerdo con el resultado. Para hacer esto se utilizan los comandos IF y THEN. En el siguiente programa puedes descubrir cómo funciona IF y THEN.

```
10 CLS
20 PRINT "TECLEA DOS NUMEROS"
30 INPUT A
40 INPUT Z
50 IF A=Z THEN PRINT "IGUAL"
60 IF A<>Z THEN PRINT "DISTINTOS"
```

Teclea dos números de uno en uno y pulsa RETURN después de cada uno de ellos.



Este símbolo compara los números para comprobar si son iguales.

Este símbolo compara los números para comprobar si son distintos.

En este programa los comandos IF y THEN hacen que el ordenador compare los números almacenados en las variables A y Z. Si los

números son idénticos, el ordenador escribe «IGUAL». Si los números son diferentes, el mensaje es «DISTINTOS».

Más formas de comparar cosas

Un ordenador puede también comparar números para descubrir si uno es mayor o menor que otro. Los dibujos de la derecha muestran los símbolos para hacer esto.



Comparaciones de palabras

Puedes usar también IF y THEN para hacer que el ordenador compare palabras. Prueba a ejecutar el siguiente programa. Los comandos IF y THEN hacen que el ordenador compare los nombres almacenados en S\$ y N\$ para comprobar si son iguales o no.

Puedes poner aquí tu nombre si lo deseas.

```
10 CLS
20 LET S$="VISCOSILLO"
30 PRINT "COMO TE LLAMAS?"
40 INPUT N$
50 IF N$<>S$ THEN PRINT "NO PUEDO AGUANTAR MONSTRUOS LLAMADOS";N$
60 IF N$=S$ THEN PRINT "HOLA ";S$;" ENCANTADO DE VERTE"
```

En algunos ordenadores este mensaje es demasiado largo para entrar en una línea, en este caso aparecerá en dos líneas como aquí.

COMO TE LLAMAS?
? ROSAURO
NO PUEDO AGUANTAR MONSTRUOS
LLAMADOS ROSAURO

Programa de edades

```
10 CLS
20 LET A=11
30 PRINT "TENGO ";A;" AÑOS"
40 PRINT "CUANTOS AÑOS TIENES"
50 INPUT B
60 IF A=B THEN PRINT "TENEMOS LA MISMA EDAD"
70 IF A>B THEN PRINT "SOY MAYOR QUE TU"
80 IF A<B THEN PRINT "SOY MENOR QUE TU"
```

Pon aquí tu edad en lugar de 11.

TENGO 11 AÑOS
CUANTOS AÑOS TIENES?
? 10
SOY MAYOR QUE TU

Cuando teclees este programa, pon tu propia edad en la línea 20. Deja a un amigo ejecutar el

programa. El ordenador descubrirá si tu amigo es mayor, menor o tiene tu misma edad.

Programas con IF y THEN

Puedes hacer que tu ordenador realice todo tipo de cosas después de un comando IF y THEN. Por ejemplo, puedes decirle que salte a otra línea del programa usando el comando GOTO. También puedes hacer que se pare el programa.

Uso de GOTO

```
10 CLS
20 LET C=17
30 PRINT "ADIVINA CUANTOS PASTELES"
40 PRINT "SE HA COMIDO VISCOSILLO?"
50 INPUT G
60 IF G<>C THEN GOTO 50
70 PRINT "LO ADIVINASTE!"
```

Esta línea hace regresar al ordenador a la línea 50 si tu respuesta es equivocada.

El ordenador ejecuta esta línea si tu respuesta es correcta.

En este programa tienes que adivinar cuántos pasteles se ha comido Viscosillo. Si tu respuesta

es equivocada, el comando GOTO hace que el ordenador regrese a la línea 50 volviéndote a preguntar.

El ordenador continuará preguntándote hasta que tu respuesta sea correcta.

Cómo mejorar el programa

El programa del pastel puede resultar más claro si el ordenador te indica que tu respuesta fue equivocada. En la mayor parte de los ordenadores puedes hacer esto poniendo un comando PRINT después de THEN en la línea 60.

Prueba esta versión de la línea 60. Comprueba la diferencia que introduce en el programa.

```
60 IF G<>C THEN PRINT "FALLASTE":GOTO 50
```

PRINT y GOTO deben estar ambos en la misma línea de IF y THEN. Separa ambos con dos puntos.

Uso de IF y THEN con STOP

1

```
10 CLS
20 LET P$="JUDIAS COCIDAS"
30 PRINT "CUAL ES LA CLAVE?"
40 INPUT A$
50 IF A$<>P$ THEN STOP
60 PRINT "CORRECTO"
70 PRINT "BIENVENIDO, AMIGO"
```

El ordenador sólo escribirá estas líneas si tecleas la clave correcta.

Esta es la clave. Se almacena en la variable P\$.

2

```
CUAL ES LA CLAVE?
? PASTELES CON CREMA

STOP AT LINE 50
```


Este mensaje te indica que el programa se ha parado.

Este programa de clave usa el comando STOP después de IF y THEN. Este le indica al ordenador

que pare el programa si tecleas una clave equivocada. La mayor parte de los ordenadores escriben

un mensaje cuando paran un programa. Esto te permite saber en qué línea se paró el programa.

Máquina matemática



Prueba este programa. Cuando lo ejecutes introduce dos números en el ordenador. Luego indícale si quieres sumar o multiplicar

Recuerda, puedes poner dos comandos en una línea si los separas con dos puntos.

```
10 CLS
20 PRINT "DAME DOS NUMEROS"
30 INPUT X:INPUT Y
40 PRINT "DEBO SUMAR"
50 PRINT "O MULTIPLICAR?"
60 PRINT "(TECLEA S O M)"
70 INPUT R$
80 IF R$="S" THEN LET Z=X+Y
90 IF R$="M" THEN LET Z=X*Y
100 PRINT,"LA RESPUESTA ES ";Z
```

La respuesta se almacena en Z.

tecleando S o M. Las líneas 80 y 90 usan IF y THEN para que el ordenador realice el cálculo solicitado.

El comando LET indica al ordenador que realice la operación y almacene la respuesta en Z.

Bucles

Un bucle hace que un ordenador realice algo, como escribir un mensaje, una y otra vez. Para hacer un bucle en un programa usa los comandos FOR, TO y NEXT.

Este es un programa con bucle. Hace que el ordenador escriba cinco veces un mensaje.

Puedes usar un bucle para repetir un mensaje como éste.

HOLA UGLYMUG
HOLA UGLYMUG
HOLA UGLYMUG
HOLA UGLYMUG
HOLA UGLYMUG

Bucle Uglymug

```
10 CLS  
20 FOR K=1 TO 5  
30 PRINT "HOLA UGLYMUG"  
40 NEXT K
```

Este es el comando PRINT en el interior del bucle.

Este es el comienzo del bucle.

Este es el fin del bucle.

El bucle en este programa comienza con el comando FOR, TO en la línea 20. Este indica al

ordenador cuántas veces repetir el comando PRINT. El comando NEXT, en la línea 40, le dice al


ordenador que vuelva atrás y repita el comando PRINT otra vez.

Cómo funciona un bucle

Un bucle es parecido a una carrera en la que el corredor tiene que dar vueltas a la pista varias veces. Cada vez que el ordenador realiza la instrucción del bucle, es como correr una vuelta de la carrera.

A J se le llama variable del bucle.

```
10 CLS
20 FOR J=1 TO 6
30 PRINT "HOLA"
40 NEXT J
```



En este programa, el ordenador usa la variable J para contar las vueltas.

Al principio J vale 1. Cada vez que el ordenador pasa por el bucle, J se

incrementa en 1 hasta que alcanza el valor 6; entonces se termina el bucle.

Interior de un bucle

```
10 CLS
20 FOR L=1 TO 20
30 PRINT "VUELTA"; L
40 NEXT L
```

Este programa te permite observar el interior de un bucle, puedes ver cómo el ordenador cuenta las vueltas.



Rompecabezas bucle

```
10 CLS
20 FOR J= TO
30 PRINT
40 NEXT J
```

Teclea los números que faltan en esta línea.

Teclea aquí tu nombre, entre comillas.

Puedes completar este programa de forma que el ordenador escriba tu nombre 10 veces. Completa los números en la línea 20 y pon tu nombre entre comillas después del comando PRINT, en la línea 30.

Aquí hay más formas de usar bucles en programas.



```
Programa contador
10 CLS
20 PRINT "HASTA QUE NUMERO"
30 PRINT "DEBO CONTAR?"
40 INPUT N
50 FOR K=1 TO N
60 PRINT K
70 NEXT K
```

Los bucles son muy útiles para hacer que un ordenador cuente. Prueba este programa que hace que el ordenador cuente hasta el

número que tú quieras. Cuando ejecutes el programa teclea el número hasta donde quieres que el ordenador cuente.

Diferentes maneras de contar

Esta línea indica al ordenador que cuente de 0 a 30 de 5 en 5.

```
10 CLS
20 FOR L=0 TO 30 STEP 5
30 PRINT L
40 NEXT L
```



Prueba a cambiar el número que sigue a STEP.

Puedes modificar la forma de contar del ordenador. Por ejemplo, puedes hacer que cuente a saltos. Para hacer esto usa el comando

STEP seguido de un número. El programa de arriba hace que el ordenador cuente a saltos de 5. Prueba a ejecutarlo.

Cuenta atrás

```
10 CLS
20 PRINT "CUENTA ATRAS"
30 FOR J=10 TO 1 STEP -1
40 PRINT J
50 NEXT J
60 PRINT "EXPLOSION!"
```

STEP con un número negativo hace que el ordenador cuente hacia atrás. Prueba este programa para ver cómo funciona.

Cómo hacer esperar al ordenador

1

```
10 CLS
20 PRINT "HOLA A TODOS"
30 FOR K=1 TO 1000
40 NEXT K
50 PRINT "HOLA OTRA VEZ"
```

Este es el bucle de retardo. Hace que el ordenador cuente de 1 a 1.000 antes de que continúe ejecutando el resto del programa.

Un bucle sin instrucciones dentro hace que el ordenador realice una pausa momentánea. Se denomina bucle de retardo. En este programa, el bucle de las líneas 30 y 40 hace que efectúe una pausa antes de escribir el segundo mensaje.

2

```
30 FOR K=1 TO 500
30 FOR K=1 TO 2000
```

Un número más pequeño hace una pausa más corta.

Un número mayor hace una pausa más larga.

Puedes alterar la duración de la pausa cambiando el segundo número del comando FOR, TO. Repite el programa anterior alargando y acortando la pausa.

Uso de la variable del bucle

1

Cada vez que el ordenador pasa por el bucle multiplica 9 por K y escribe la respuesta.

```
10 CLS
20 FOR K=1 TO 12
30 LET A=K*9
40 PRINT K;" x 9 =" ;A
50 NEXT K
```

Este programa hace que el ordenador escriba la tabla del 9. La variable del bucle, K, se usa para contar en el bucle y para hacer la operación en su interior.


2

Esto aparecerá en la pantalla cuando ejecutes el programa.

```
10 CLS
20 FOR J=1 TO 10
30 PRINT TAB(J);"HOLA"
40 NEXT J
```

En este programa el ordenador usa la variable del bucle, J, como número para el comando PRINT TAB.

Esto significa que cada vez que el ordenador pasa por el bucle, deja un espacio más antes de la palabra HOLA.



Números aleatorios

Cuando tiras los dados es bastante difícil adivinar o predecir el número que saldrá. Los números impredecibles como éstos se denominan números aleatorios.

A la derecha puedes descubrir cómo hacer que un ordenador produzca números aleatorios. El comando es bastante complicado, pero resulta muy útil en toda clase de programas.



1

Mira la página 33 para encontrar el comando RND usado por tu ordenador.

`INT(RND(1))`

En el recuadro de la derecha puedes descubrir qué números poner aquí.

Para que tu ordenador produzca un número aleatorio usa los comandos INT y RND(1) (o RND o RND(0) en algunos ordenadores) seguido de otros números.

Escritura de números aleatorios

```
10 CLS
20 PRINT "AQUI HAY UN NUMERO"
30 PRINT "ENTRE 1 Y 20"
40 PRINT INT(RND(1)*20+1)
```



Recuerda cambiar esto, si es necesario, en tu ordenador.

Hay 20 números para elegir, comenzando por el 1.




AQUI HAY UN NUMERO
ENTRE 1 Y 20
17

El ordenador puede escoger el mismo número en algún caso.

Este es un sencillo programa de prueba con números aleatorios. La línea 40 le indica al ordenador que

escoja un número entre 1 y 20 y lo escriba en la pantalla. Ejecuta el programa varias veces para

practicar. El ordenador escogerá un número distinto cada vez.



Para completar la instrucción necesitas multiplicar por el total de números de entre los que el ordenador puede escoger y sumar el número más bajo de los que puede escoger. Por ejemplo, para que el ordenador escoja entre 1 y 6, como un dado, necesitas la instrucción de arriba. Asegúrate que usas el comando RND correcto para tu ordenador.

Si introduces en el ordenador el comando mostrado en el recuadro 2, éste escoge un número a ciegas entre 1 y 6.

Programa lotería

HOLA MONSTRUOS EL AFORTUNADO GANADOR ES EL NUMERO 19

Recuerda cambiar esto si te es necesario.

Debes completar los números que faltan aquí.

```

10 CLS
20 LET W=INT(RND(1)*  + )
30 PRINT "HOLA MONSTRUOS"
40 PRINT "EL AFORTUNADO GANADOR"
50 PRINT "ES EL NUMERO ";W
  
```

La panda de los monstruos está haciendo una lotería. Han vendido 30 papeletas, numeradas del 1 al 30, y quieren que el ordenador escoja el número ganador. El programa de la derecha hace que el ordenador realice esto, pero no está

completo. Prueba a completar los números que faltan en la línea 20. Esta es la línea que indica al ordenador que elija un número aleatorio entre 1 y 30. Encontrarás la respuesta en la página 45 del libro.

Cómo depurar programas

Depurar un programa significa corregir los errores existentes. A continuación puedes encontrar cómo depurar programas.

En la parte inferior de la página hay algunas sugerencias para localizar «bugs».

1 SYNTAX ERROR EN LINEA 20

«Syntax error» significa que hay algo erróneo en un comando BASIC.



2

```
LIST 10 CLS
20 PRUNT "HOLA"
30 PRINT "MONSTRUO"
```

¿Puedes localizar el error en este programa?




3

El comando PRINT está escrito erróneamente.

```
LIST
10 CLS
20 PRUNT "HOLA"
30 PRINT "MONSTRUOS"
```

20 PRINT "HOLA"

Línea reescrita correctamente.



Tu ordenador puede escribir un mensaje de error para ayudarte a encontrar el error. Mira tu manual para averiguar lo que significa el mensaje de error.

Para encontrar un error, teclea LIST y pulsa RETURN. Esto escribe las líneas del programa en la pantalla. Comprueba cada línea del programa.

Cuando localices un error, corrígelo tecleando la línea completa de nuevo, comenzando con el número de línea.

Sugerencias para localizar errores

Los errores son bastante fáciles corregir, pero a menudo son difíciles de localizar. Incluso el más pequeño error puede parar la ejecución de un programa. A la derecha hay algunas sugerencias a tener en cuenta.

Este programa contiene algunos errores usuales.

Se ha usado la letra O en vez del número 0.

Comando BASIC erróneamente tecleado

Faltan comillas.

```
10 PRINT COMO TE LLAMAS?
20 INOUT AS
30 PRINT "HOLA" AS
40 PRINT "ADIOS ";AS
```

Falta punto y coma.

Tabla de comandos

La tabla de más abajo muestra los comandos que usan diferentes ordenadores para borrar la pantalla y escoger números aleatorios. Al final

de esta página puedes encontrar los nombres para las teclas RETURN y DELETE en diferentes ordenadores.

Necesitas completar los números en este comando según se dijo en la página 31.

| | Borrar pantalla | Número aleatorio |
|-----------------------|------------------|------------------|
| MSX | CLS | INT(RND(1)* +) |
| VIC 20 y Commodore 64 | PRINT CHR\$(147) | INT(RND(1)* +) |
| Apple | HOME | INT(RND(1)* +) |
| Spectrum | CLS | INT(RND* +) |
| Dragón | CLS | INT(RND)0)RETURN |

Tabla de teclas

| | RETURN | DELETE |
|-----------------------|--------|------------------|
| MSX | RETURN | DEL |
| VIC 20 y Commodore 64 | RETURN | INST DEL |
| Apple | RETURN | ← |
| Spectrum | ENTER | CAPS SHIF Ø _ |
| Dragón | ENTER | ← |

Pulsa ambas teclas a la vez

Programas y rompecabezas

En esta página y las siguientes encontrarás programas de prueba y rompecabezas para resolver. Todos usan los comandos BASIC

explicados en este libro. Si tienes problemáticas con alguno de los rompecabezas, mira la respuesta en las páginas 45-47.

1 Palabras y espacios

```
10 CLS
20 PRINT "ROJO";"ROSA"
30 PRINT "NEGRO";"CARBON"
40 PRINT "BLANCO";"NIEVE"
```

Prueba este programa. Hace que el ordenador escriba palabras unas junto a otras en la pantalla.

2

```
ROJO ROSA
NEGRO CARBON
BLANCO NIEVE
```

Trata de cambiar las líneas PRINT para introducir un espacio entre las palabras, como se muestra arriba.

3

```
ROJO      ROSA
NEGRO     CARBON
BLANCO    NIEVE
```

¿Puedes cambiar de nuevo las líneas PRINT para hacer que el ordenador coloque las palabras en dos columnas?

Líneas de estrellas

1

```
10 CLS
20 FOR K=1 TO 10
30 PRINT TAB(3);"*"
40 NEXT K
```

Ejecuta este programa que hace que tu ordenador escriba una línea de estrellas en un lado de la pantalla.

2

```
*
*
*
*
*
*
*
*
*
*
```

Cambiando el número que sigue a TAB, ¿puedes poner las estrellas en el centro de la pantalla?

3

```
*****
```



Sugerencia: usa el punto y coma para hacer que el ordenador permanezca en la misma línea.

¿Puedes cambiar de nuevo el programa para hacer que una línea de estrellas como éstas cruce la pantalla?

Rompecabezas de invitación a fiesta

Si quieres, puedes poner tu propio nombre en la invitación y cambiar el contenido.

```
10 CLS
20 PRINT "*****"
30 PRINT "POR FAVOR VEN"
40 PRINT "A MI"
50 PRINT "FIESTA DE CUMPLEAÑOS"
60 PRINT "FECHA 19 JULIO"
70 PRINT "HORA 3.00 PM"
80 PRINT "LUGAR MI CASA"
90 PRINT "CARIÑOSAMENTE"
100 PRINT "UGLYMUG"
110 PRINT "*****"
```

```
*****

POR FAVOR VEN
A MI
FIESTA DE CUMPLEAÑOS

FECHA      19 JULIO
HORA       3.00 PM
LUGAR      MI CASA

CARIÑOSAMENTE
UGLYMUG

*****
```

Este programa escribe una invitación a una fiesta en la izquierda de la pantalla. ¿Puedes centrar la invitación en la pantalla, según se muestra a la derecha?

Necesitas cambiar el comando PRINT por PRINT , TAB y añadir algunas líneas PRINT vacías para poner las líneas en blanco.

4

```
*****
*
*
*
*
*
*
*
*
*
```

Prueba a escribir un programa con dos bucles para escribir estrellas en forma de T.

5

```
10 CLS
20 FOR K=1 TO 10
30 PRINT TAB(K);"*"
40 NEXT K
```

Este programa usa PRINT TAB seguido de una variable de bucle para dibujar una línea de estrellas en diagonal. Cada vez

6

```
*
*
*
*
*
*
*
*
*
```

que el ordenador ejecuta el bucle, aparece la estrella un espacio más allá a la izquierda de la pantalla.

Programas de cálculos

Los programas de estas dos páginas hacen que el ordenador realice cálculos. Pruébalos y mira si puedes resolver los rompecabezas.

Tiempo de tele

```
10 CLS
20 PRINT "CUANTAS HORAS"
30 PRINT "VES LA TELE"
40 PRINT "CADA SEMANA?"
50 INPUT X
60 LET Y=X*52
70 PRINT "ESTO SIGNIFICA QUE VES"
80 PRINT Y;" HORAS LA TELE"
90 PRINT "CADA AÑO!"
```

CUANTAS HORAS
VES LA TELE
CADA SEMANA?
? 6
ESTO SIGNIFICA QUE VES
312 HORAS LA TELE
CADA AÑO!

Este programa calcula el número de horas que ves la tele cada año. Cuando ejecutes el programa teclea el número de horas que ves la tele a la semana. El ordenador lo multiplica por 52 (número de semanas en un año). La línea 60 realiza los cálculos.

Rompecabezas de cálculos

1

```
CUANTOS HELADOS  
COMES  
CADA SEMANA?  
? 2  
ESTO SIGNIFICA QUE COMES  
104 HELADOS  
CADA AÑO!
```

2

```
CUANTOS CARAMELOS  
COMES CADA DIA?  
? 2  
ESTO SIGNIFICA QUE COMES  
730 CARAMELOS  
CADA AÑO!
```



Necesitas cambiar los
cálculos de la línea 60.

¿Puedes adaptar el programa anterior para calcular cuántos helados comes cada año?

Ahora prueba a cambiar el programa de nuevo para hacer que calcule cuántos caramelos comes cada año, partiendo del número de caramelos que comes al día, según se muestra arriba.

Programa de promedios

```
10 CLS
20 PRINT "CUANTAS EDADES"
30 PRINT "QUIERES PROMEDIAR?"
40 INPUT N
50 LET T=0
60 FOR K=1 TO N
70 PRINT "CUAL ES LA EDAD ";K;"?"
80 INPUT X
90 LET T=T+X
100 NEXT K
110 LET A=T/N
120 PRINT "EL PROMEDIO ES ";A
```

Debes hacer que el ordenador pregunte por el tiempo de cada viaje en minutos.



Rompecabezas

```
CUANTOS VIAJES
QUIERES PROMEDIAR
? 5
TECLEA TIEMPO EN MINUTOS
VIAJE 1
? 21
VIAJE 2
? 25
VIAJE 3
? 31
VIAJE 4
? 19
VIAJE 5
? 29
EL PROMEDIO DE TUS
VIAJES ES 25 MINUTOS
```

El programa de arriba calcula la edad media de un grupo de personas. Usalo para averiguar la edad media de tu familia o amigos. Cuando lo ejecutes, teclea el número de edades que quieres

promediar y luego las edades. Cuando tecleas cada edad, el ordenador las suma para calcular la edad total (T). Luego divide T por el número de edades (N) para obtener el promedio.

Puedes modificar fácilmente el programa de la izquierda para calcular diferentes promedios. Prueba a hacerlo para calcular el tiempo medio que tardas en ir al colegio.

Operaciones

1

```
10 CLS
20 PRINT "TECLEA DOS"
30 PRINT "NUMEROS"
40 INPUT A
50 INPUT B
60 PRINT A;"x";B;"=";
70 PRINT A x B
```

Este programa hace que el ordenador multiplique dos números cualesquiera, pero contiene un error. ¿Puedes localizarlo?

2

```
TECLEA DOS NUMEROS
? 4
? 2
4 + 2 = 6
4 - 2 = 2
4 x 2 = 8
```

Prueba a hacer un programa para hacer que el ordenador sume, reste y multiplique dos números cualesquiera.

3

```
TECLEA UN NUMERO
? 72
72 x 5 = 360
```

¿Puedes escribir un programa para hacer que el ordenador multiplique cualquier número por 5?

Rompecabezas con bucles

Teclea estos programas y rompecabezas para practicar el uso de bucles

Programa acertijo

```
10 CLS
20 PRINT "QUE ES VERDE Y PELUDO"
30 PRINT "Y VA HACIA ARRIBA Y ABAJO?"
40 PRINT:PRINT:PRINT
50 FOR K=1 TO 2000:NEXT K
60 PRINT "UNA GROSELLA"
70 PRINT "EN UN ASCENSOR!!"
```

En este programa el bucle de retardo de la línea 50 hace una

pausa en el ordenador antes de escribir la respuesta.

QUE ES VERDE Y PELUDO
Y VA HACIA ARRIBA Y ABAJO?

UNA GROSELLA
EN UN ASCENSOR!!

Puedes poner las dos partes
del bucle en una línea.
Sepáralas con dos puntos.

Prueba a ejecutar el programa
para ver cómo funciona.

Rompecabezas con bucles de retardo

1

Estos son bucles de retardo.

```
10 CLS
20 FOR K=1 TO 100
30 PRINT "HOLA"
40 FOR L=1 TO 20:NEXT L
50 CLS
60 FOR L=1 TO 20:NEXT L
70 NEXT K
```

Este programa coloca un mensaje intermitente en la pantalla, pero esta intermitencia es demasiado rápida y apenas puedes leerlo. ¿Puedes resolver este problema?

2

```
10 CLS
20 FOR L=1 TO 60
30 PRINT L
40 FOR J=1 TO 50:NEXT J
50 NEXT L
```

Usa un reloj como ayuda.

En este programa el ordenador cuenta de 1 a 60. Prueba a cambiar la duración del retardo para hacer que el ordenador cuente con pausas de un segundo. Prueba con retardos de distintas duraciones.

Secuencias de números

```
1  10 CLS
   20 LET R=INT(RND(1)*10+1)
   30 PRINT "AQUI HAY TRES NUMEROS"
   40 FOR J=2 TO 4
   50 PRINT J*R
   60 NEXT J
   70 PRINT "CUAL ES EL"
   80 PRINT "SIGUIENTE"
   90 LET N=5*R
  100 INPUT X
  110 IF X=N THEN PRINT "SI, BIEN HECHO"
  120 IF X<>N THEN PRINT "NO, ES ";N
```

El programa de arriba es un juego de números. Usa un número aleatorio y un bucle para

producir una secuencia de tres números. Tienes que adivinar el siguiente número de la secuencia.

2

```
AQUI HAY TRES NUMEROS
6
9
12
CUAL ES EL
SIGUIENTE
? 15
SI, BIEN HECHO
```

Esto aparecerá en tu pantalla cuando ejecutes el programa.

3

```
AQUI HAY TRES NUMEROS
24
18
12
CUAL ES EL
SIGUIENTE
? 6
SI, BIEN HECHO
```

Sugerencia: necesitas un bucle seguido de STEP con un número negativo.

¿Puedes cambiar el juego, de forma que la secuencia de números sea descendente? Necesitas alterar el bucle y los cálculos de la línea 90.

Rompecabezas para contar

```
0
100
200
300
400
500
```

¿Puedes escribir un programa que cuente de 0 a 500 de 100 en 100?

Rompecabezas pantalla


Sugerencia: necesitas poner un comando PRINT dentro de un bucle.

ESTA ES LA PARTE INFERIOR DE LA PANTALLA!

Prueba a hacer que el ordenador escriba un mensaje como éste en la parte inferior de la pantalla.

Rompecabezas con números

Prueba con los rompecabezas de la derecha. Te ayudarán a practicar el comando de números aleatorios en tu ordenador. Luego trata de completar el comando de números aleatorios del programa de juegos de más abajo.



1

EL NUMERO ALEATORIO
ES 18

¿Puedes hacer un programa para que el ordenador escoja un número aleatorio entre 10 y 20 y lo escriba en la pantalla como aquí se indica?


2

AQUI HAY CINCO NUMEROS
10
13
19
17
14

Prueba ahora a escribir un programa para hacer que el ordenador escriba cinco números aleatorios.


Juego de dados

TU TIRADA
? 5
MI TIRADA
? 3
TU GANAS



Para jugar a este juego necesitas un dado. Tira el dado y teclea el número que obtienes.

```
10 CLS
20 PRINT "TU TIRADA"
30 INPUT X
40 PRINT "MI TIRADA"
50 LET R=
60 PRINT R
70 IF X=R THEN PRINT "EMPATE"
80 IF X>R THEN PRINT "TU GANAS"
90 IF X<R THEN PRINT "YO GANO"
```



Luego el ordenador «tira» escogiendo un número entre 1 y 6. Antes de ejecutar el programa

debes poner el comando de números aleatorios en la línea 50.

Juego de adivinar números

```
10 CLS
20 LET R=
30 PRINT "ESTOY PENSANDO EN UN"
40 PRINT "NUMERO ENTRE"
50 PRINT "1 Y 25"
60 PRINT "ADIVINA CUAL ES"
70 INPUT G
80 IF G=R THEN GOTO 120
90 IF G<R THEN PRINT "DEMASIADO PEQUEÑO"
100 IF G>R THEN PRINT "DEMASIADO GRANDE"
110 GOTO 70
120 PRINT "SI, ACERTASTE"
```

En este juego, el ordenador elige un número y tú debes adivinarlo. Cada vez que lo intentas, el ordenador te indica si tu número

de demasiado pequeño o demasiado grande. El juego no termina hasta que adivinas el número. Antes de ejecutar el

```
ESTOY PENSANDO EN UN
NUMERO ENTRE
1 Y 25
ADIVINA CUAL ES.
? 4
DEMASIADO PEQUEÑO
? 6
DEMASIADO GRANDE
? 5
SI, ACERTASTE
```

programa, coloca el comando de números aleatorios en la línea 20 para que el ordenador escoja un número entre 1 y 25.


Rompecabezas

1

```
ESTOY PENSANDO EN UN
NUMERO ENTRE
1 Y 25
ADIVINA CUAL ES
? 8
DEMASIADO GRANDE
? 7
DEMASIADO GRANDE
? 3
SI, ACERTASTE
HAS HECHO 3 INTENTOS
```

2

15 LET C=0



Tienes que descubrir qué números de línea deben tener las líneas que faltan.

¿Puedes mejorar el juego de forma que el ordenador cuente el número de intentos? Necesitarás una nueva variable C. Primero añade la línea 15 mostrada arriba, para almacenar el cero

en C. Luego debes añadir una línea que sume 1 en cada vez que haces un intento. Finalmente, añade una línea PRINT para escribir el número de intentos que has hecho.

Rompecabezas para localización de errores


En los programas de estas dos páginas hay errores deliberados. Algunos son errores que paran la ejecución del programa. Otros provocan que el programa haga tonterías. Comprueba si puedes detectarlos y corregirlos para que los programas funcionen correctamente.

Contador

```
1
10 CLS
20 PRINT "PRIMER NUMERO?"
30 INPUT A
40 PRINT "ULTIMO NUMERO?"
50 INPUT B
60 FOR K=B TO A
70 PRINT K
80 NEXT K
```



2 PRIMER NUMERO?
? 12
ULTIMO NUMERO?
? 15
12
13
14
15



El programa de la izquierda hace que el ordenador cuente. Tú eliges el número donde empieza a contar y hasta dónde lo hace. ¿Puedes localizar el error en el programa?

Tabla de multiplicar

```
1
10 CLS
20 PRINT "TABLA DE MULTIPLICAR"
30 PRINT:PRINT
40 PRINT "QUE TABLA QUIERES?"
50 INPUT T
60 FOR J=1 TO 12
70 LET A=J*T
80 PRINT J;" x ";T;" = ";A
90 NEXT Z
```




2
TABLA DE MULTIPLICAR
QUE TABLA QUIERES?
? 6
1 x 6 = 6
2 x 6 = 12
3 x 6 = 18
4 x 6 = 24
5 x 6 = 30
6 x 6 = 36
7 x 6 = 42
8 x 6 = 48
9 x 6 = 54
10 x 6 = 60
11 x 6 = 66
12 x 6 = 72

Si ejecutas este programa, no funcionará. ¿Puedes encontrar el error? El mensaje de error del ordenador puede ayudarte a localizarlo.

Esto aparecerá en la pantalla cuando el programa funcione correctamente.

Cara feliz o triste

Para averiguar el error en el programa de abajo, trata de ejecutarlo. Debería hacer que el ordenador escribiera una cara feliz cuando teclearas FELIZ, y una cara triste cuando teclearas TRISTE. Cuando hayas localizado el error, prueba a corregir el programa.



```


10 CLS
20 PRINT "TE SIENTES"
30 PRINT "FELIZ O TRISTE?"
40 INPUT A$
50 PRINT:PRINT:PRINT
60 PRINT "□////////□"
70 PRINT "!□□□□□□□!"
80 IF A$="TRISTE" THEN PRINT "!□□□□□□!"
90 IF A$="FELIZ" THEN PRINT "!□-□□□-□!"
100 PRINT "!□□□,□□□!"
110 IF A$="TRISTE" THEN PRINT "!□:□□□:□!"
120 IF A$="FELIZ" THEN PRINT "!□□□□□□□!"
130 PRINT "!□:::□!"
140 IF A$="TRISTE" THEN PRINT "!□□□□□□□!"
150 IF A$="FELIZ" THEN PRINT "!□:□□□:□!"
160 PRINT "□!□□□□□!□"
170 PRINT "□XXXXXXXX□"
180 PRINT:PRINT
190 IF A$="FELIZ" THEN PRINT "CARA FELIZ"
200 IF A$="TRISTE" THEN PRINT "CARA TRISTE"
    
```



Este símbolo indica teclear un espacio. Cada vez que lo veas pulsa una vez la barra de espacios.



Estas líneas hacen que el ordenador dibuje los ojos.



Las líneas 110 a 150 hacen que el ordenador dibuje la boca.



Respuestas a los rompecabezas

Para encontrar la respuesta a un rompecabezas, mira el número de la página y el nombre del rompecabezas.

En alguno de ellos puedes encontrar que tu programa es ligeramente distinto a la respuesta. Esto no es importante siempre que tu programa funcione correctamente.

Página 12 Programa para dibujar una cara

```
10 CLS
20 PRINT "□□***"
30 PRINT "□□*****"
40 PRINT "□*****"
50 PRINT "□1□□□□1"
60 PRINT "□1□o□o□1"
70 PRINT "(□□,□□)"
80 PRINT "□1□□□□1"
90 PRINT "□1□□0□1"
100 PRINT "-----"
```



Este programa dibuja una cara como la mostrada.

Página 14 Rompecabezas LET

```
LET C=11
LET G=19
LET Z=30
```

Estos comandos LET asignan los números a las variables.

Página 15 Rompecabezas PRINT

```
50 PRINT A$
60 PRINT B$
70 PRINT C$
```

Para ver las palabras almacenadas en A\$, B\$ y C\$, coloca los comandos PRINT seguidos de los nombres de las variables, según se muestra.

Página 19 Calculadora

```
30 INPUT A
50 INPUT B
```

Estos son los comandos de las líneas 30 y 50. Indican al ordenador que almacene los números que teclees, en las variables A y B.

Programa mensaje

```
90 PRINT A$
110 PRINT B$
```

Para que el ordenador escriba los nombres almacenados en A\$ y B\$, coloca los nombres de las variables después de los comandos PRINT.

Página 21 Rompecabezas de programa de cumpleaños

```
40 PRINT TAB(13);"FELIZ CUMPLEAÑOS"
50 PRINT TAB(18);"A TI"
60 PRINT TAB(17);"*****"
```

Para centrar las palabras en la pantalla, coloca los números en el paréntesis que sigue a TAB. Los números varían de unos ordenadores a otros. Los mostrados corresponden a la mayor parte de los ordenadores excepto Spectrum y VIC 20. Debes usar números más pequeños en dichos ordenadores.

Página 27 Rompecabezas bucle

```
20 FOR J=1 TO 10
30 PRINT "ROBERTO EL TUERTO"
```

Debes poner los números 1 y 10 en la línea 20 y añadir tu nombre al comando PRINT de la línea 30, según se muestra.

Página 31 Programa lotería

```
20 LET W=INT(RND(1)*30+1)
```

Este es el comando completo de números aleatorios. Puede que tengas que quitar el (1) o poner el (0) en tu ordenador.

No importa a qué lado del punto y coma teclees el espacio.

Página 34 Palabras y espacios

```
2 20 PRINT "ROJO "; "ROSA"
30 PRINT "NEGRO "; "CARBON"
40 PRINT "BLANCO"; "NIEVE"
```

Para dejar espacios entre palabras, teclea un espacio dentro de las comillas a uno de los lados del punto y coma.

```
3 20 PRINT "ROJO","ROSA"
30 PRINT "NEGRO","CARBON"
40 PRINT "BLANCO","NIEVE"
```

Para poner los nombres en columnas, cambia el punto y coma por coma. No necesitas dejar espacios dentro de las comillas.

Líneas de estrellas

2 Para hacer una línea de estrellas vertical en el centro de la pantalla, pon un número elevado después de TAB. En la mayor parte de los ordenadores es 18. En el Spectrum es 15 y en el VIC 20 es 10.

```
3 30 PRINT "*";
```

Para escribir una línea de estrellas horizontal, quita el comando TAB de la línea 30 y pon un punto y coma después de la estrella.

30 es el número de papeletas vendidas.

```
4 10 CLS
20 FOR K=1 TO 10
30 PRINT "*";
40 NEXT K
50 FOR K=1 TO 10
60 PRINT TAB(5); "*"
70 NEXT K
```

Este es el programa para hacer las líneas de estrellas en forma de T. Usa los bucles de los rompecabezas 2 y 3.

Página 35 Rompecabezas de invitación a una fiesta

Para centrar las líneas, reemplaza los comandos PRINT por PRINT TAB. El número después de TAB variará dependiendo de tu ordenador.

Página 36 Rompecabezas de cálculos

```
1 20 PRINT "CUANTOS HELADOS"
30 PRINT "ESTO SIGNIFICA QUE COMES"
80 PRINT Y;" HELADOS"
```

Debes cambiar las palabras en las líneas PRINT según se muestra

```
2 20 PRINT "CUANTOS CARAMELOS"
30 PRINT "COMES CADA DIA?"
60 LET Y=X*365
70 PRINT "ESTO SIGNIFICA QUE COMES"
80 PRINT Y;" CARAMELOS"
```

365 es el número de días de un año.

Cambia las líneas PRINT según se muestra. No necesitas la línea 40. Para quitarla teclea 40 y pulsa RETURN. En la línea 60 se multiplica X por 365.

Página 37

Rompecabezas de promedios

```
20 PRINT "CUANTOS VIAJES"
30 PRINT "QUIERES PROMEDIAR?"
45 PRINT "TECLEA TIEMPO EN MINUTOS"
70 PRINT "VIAJE ";K
120 PRINT "EL PROMEDIO DE TUS"
130 PRINT "VIAJES ES ";A;" MINUTOS"
```

Para modificar el programa de promedios cambia las palabras de las líneas PRINT. Deberás también añadir un comando PRINT (línea 45) para preguntar por la duración del viaje en minutos.

Operaciones

1 70 PRINT A*B

La operación en la línea 70 no es correcta. Deberás usar el signo de multiplicación, *, en lugar de x.

```
2 10 CLS
20 PRINT "TECLEA DOS NUMEROS"
30 INPUT X:INPUT Y
40 PRINT X;" + ";Y;" = ";X+Y
50 PRINT X;" - ";Y;" = ";X-Y
60 PRINT X;" x ";Y;" = ";X*Y
```

Este programa hace que el ordenador sume, reste y multiplique dos números cualesquiera.

```
3 10 CLS
20 PRINT "TECLEA UN NUMERO"
30 INPUT N
40 PRINT N;" x 5 = ";N*5
```

Este programa hace que el ordenador multiplique cualquier número por 5

Página 38

Rompecabezas con bucle de retardo

1 Debes hacer mayores los bucles de retardo de las líneas 40 y 60. Para hacerlo, coloca un número mayor del comando TO.

2

| | |
|--------------|---------------------------|
| MSX | 40 FOR J=1 TO 470 :NEXT J |
| Dragón | 40 FOR J=1 TO 650 :NEXT J |
| VIC 20 | 40 FOR J=1 TO 860 :NEXT J |
| Commodore 64 | 40 FOR J=1 TO 710 :NEXT J |
| Apple | 40 FOR J=1 TO 820 :NEXT J |
| Spectrum | 40 FOR J=1 TO 220 :NEXT J |

El tamaño de los bucles varía de unos ordenadores a otros. Esta tabla da los bucles de retardo para distintos ordenadores.

Página 39

Secuencias de números

```
3 40 FOR=4 TO 2 STEP-1
90 LET N=1*R
```

STEP-1 en la línea 40 hace que la secuencia de números sea descendente. Los cálculos de la línea 90 obtienen el siguiente número de la secuencia.

Rompecabezas para contar

```
10 CLS
20 FOR J=0 TO 500 STEP 100
30 PRINT J
40 NEXT J
```


Necesitas un programa con un STEP como el indicado para que el ordenador cuente de 0 a 500 en saltos de 100.

Rompecabezas pantalla

```
10 CLS
20 FOR K=1 TO 18
30 PRINT
40 NEXT K
50 PRINT "ESTA ES LA PARTE INFERIOR"
60 PRINT "DE LA PANTALLA!"
```

Este programa hace que el ordenador deje 18 líneas en blanco, de esta forma el mensaje aparece en la parte inferior de la pantalla.

En algunos ordenadores puede ser necesario cambiar el 18 por 13.



Página 40

Rompecabezas con números

```
1  10 CLS
   20 LET R=INT(RND(1)*11+10)
   30 PRINT "EL NUMERO ALEATORIO"
   40 PRINT "ES ";R
```

Necesitas un programa como éste para que el ordenador escoja un número entre 10 y 20 y lo escriba. En tu ordenador puede que necesites cambiar RND(1) por RND o RND(0)

```
2  10 CLS
   20 PRINT "AQUI HAY CINCO NUMEROS"
   30 FOR L=1 TO 5
   40 LET R=INT(RND(1)*11+10)
   50 PRINT R
   60 NEXT L
```

El número aleatorio y el comando PRINT están dentro de un bucle. Cada vez que el ordenador pasa por el bucle escoge otro número y lo escribe en la pantalla.

Juego de dados

```
50 LET R=INT(RND(1)*6+1)
```

Debes poner este comando de números aleatorios en la línea 50. En algunos ordenadores necesitas cambiar RND(1) por RND o RND(0).

Página 41

Juego de adivinar números

```
20 LET R=INT(RND(1)*25+1)
```

Este es el comando completo de números aleatorios. Recuerda cambiar RND(1) en algunos ordenadores.

Hay 11 números para elegir entre 10 y 20. El menor es 10.



Rompecabezas

```
75 LET C=C+1
130 PRINT "HAS HECHO ";C;" INTENTOS"
```

Debes añadir estas dos líneas, así como la línea 15. La línea 75 suma 1 al número en C cada vez que haces un intento. La línea 130 escribe el número total de intentos.

Página 42

Contador

Las variables A y B están cambiadas en el comando FOR, TO. La línea 60 debe ser:

```
60 FOR K=A TO B
```

Tabla de multiplicar

Al comando NEXT le sigue una variable equivocada. Debería estar seguido de J, la variable de bucle.

```
90 NEXT J
```

Página 43

Cara feliz o triste

Cuando tecleas FELIZ, el ordenador dibuja una cara triste y viceversa. Para corregirlo, intercambia las palabras FELIZ y TRISTE en las líneas IF y THEN según se muestra.


```
80 IF A$="FELIZ" THEN PRINT "!□□□□□!"
90 IF A$="TRISTE" THEN PRINT "!□-□□□-□!"
110 IF A$="FELIZ" THEN PRINT "!□:□□□:□!"
120 IF A$="TRISTE" THEN PRINT "!□□□□□!"
140 IF A$="FELIZ" THEN PRINT "!□□□□□!"
150 IF A$="TRISTE" THEN PRINT "!□:□□□:□!"
```


- BASIC, 4, 6
- bucles, 26-27, 28-29, 34, 35, 38-39
- bucle de retardo, 29, 38
- columnas, 21, 34
 - cómo hacer, 21
- coma, 21
- comando, 4, 6
- comparaciones, 22-23, 24-25
 - símbolos para, 22-23
- contar, hacer al ordenador, 28, 39, 42
- cursor, 6
- DELETE, 10, 33
- depurar programas, 32
- división, 13, 17, 37
- dos puntos, 21, 24, 38
- ENTER, 6, 33
- errores, 11, 16, 42-43
 - cómo corregirlos, 32
- FOR...NEXT, 26-27, 28-29
- FORTRAN, 5
- GOTO, 24
- hilera, 15
- HOME, 7, 33
- IF...THEN, 22-23, 24-25
- INPUT, 18-19
- INT, 30-31, 33
- LET, 14
- LIST, 11, 32
- LOGO, 5
- memoria, 10, 14-15, 18
- mensaje de error, 11, 32
- multiplicación, 13, 17, 25, 36, 37, 39
- NEW, 11
- números aleatorios, 30-31, 33, 39, 40-41
- operaciones, 13, 17, 25, 29, 36-37
- ordenador
 - código, 5
 - lenguajes, 4-5
- PRINT, 6-7, 12-13, 15, 16, 17, 20-21, 34
- CHR\$ (147), 7, 33
- escribir mensajes con, 20, 21, 35, 39
- hacer dibujos con, 12, 43
- TAB, 20-21, 29, 34-35
- programas, 4
- cómo modificar, 11
- cómo introducir y ejecutar, 10, 11
- escribir, 8-9
- promedios, 37
- punto y coma, 16
- resta, 36
- RETURN, 6, 10, 33
- RND, 30-31, 33
- RUN, 10
- STEP, 28, 39
- STOP, 25
- sumas, 13, 17, 19, 25
- tabla de teclas, 33
- tabla de multiplicar, 29, 42
- tortuga, 5
- variables, 14-15, 16-17, 18-19, 41
 - de bucle, 27, 29, 35

Traducido por Pedro Chamero
 Revisado por Javier David García y
 Javier Enríquez de Salamanca

Queda prohibida la reproducción total o parcial de la presente obra
 bajo cualquiera de sus formas, gráfica o audiovisual, sin la autorización
 previa y escrita del editor, excepto citas en revistas, diarios o libros,
 siempre que se mencione la procedencia de las mismas.

Título original: *Simple Basic*
 © 1984 Usborne Publishing Ltd.
 © 1985, de la edición española,
 Ediciones Generales Anaya
 Villafranca, 22. 28028 Madrid
 ISBN: 84-7525-261-3
 Depósito legal: M. 25.157/1985
 Impreso por Edime, S. A.
 Calle D, esquina a F
 Polígono Industrial de Arroyomolinos
 Móstoles (Madrid)
 Impreso en España - Printed in Spain



MI PRIMERA BIBLIOTECA INFORMATICA

He aquí una nueva serie, divertida y llena de colorido, para que los más jóvenes se inicien en el mundo apasionante de los ordenadores, de la mano de una banda de fantasmas y monstruos que nos van mostrando lo que pueden hacer los ordenadores, cómo funcionan y cómo programarlos. Los libros contienen programas elementales, cortos y divertidos que se pueden utilizar en cualquier ordenador personal.



TODO SOBRE ORDENADORES

Esta amenísima introducción explica en términos sencillos qué es un ordenador, cómo funciona y todo lo que puede hacer. El libro, lleno de divertidas ilustraciones, contiene varios programas cortos para hacer más comprensibles sus explicaciones.



BASIC FACIL

Un libro divertido y sencillo para que cualquiera aprenda con toda facilidad a programar en BASIC, ayudándose de los abundantes programas para realizar y de los rompecabezas para resolver que incluye en sus páginas, profusamente ilustradas.



EL ORDENADOR DIVERTIDO

Un ordenador puede utilizarse como si fuese un juguete, pero en tal caso resulta tan apasionante que es difícil dejarlo. En este libro se demuestra que, sin necesidad de conocimientos previos, se puede usar el ordenador como fascinante baúl de juegos.

EDICIONES GENERALES ANAYA